```
> restart:
```

Figure out stages 1 and 2 of GaussRK2

Let's approximate X1 and X2 by integrating a linear fit through (t1,F1), (t2,F2):

```
> f_:=t->a*t+b; # Linear fit
```

$$f\_ := t \mapsto a \cdot t + b \tag{1}$$

```
> t1:=(1/2-sqrt(3)/6)*dt; t2:=(1/2+sqrt(3)/6)*dt; # Gauss points
  for two-point Gauss rule
```

$$t1 := \left(\frac{1}{2} - \frac{\sqrt{3}}{6}\right) dt$$

$$t2 := \left(\frac{1}{2} + \frac{\sqrt{3}}{6}\right) dt \tag{2}$$

```
> sol:=solve({f_(t1)=F1, f_(t2)=F2},{a,b}); # Find linear fit
  through (t1,F1), (t2, F2)
```

$$sol := \left\{a = \frac{\sqrt{3}\,(-F1 + F2)}{dt},\, b = \frac{\left(1+\sqrt{3}\right)\left(F1 - 2\,F2 + \sqrt{3}\,F2\right)}{2}\right\} \tag{3}$$

```
> linear_fit:=simplify(eval(f_(x),sol));
```

$$linear\_fit := \frac{-2\,(F1 - F2)\left(x - \frac{dt}{2}\right)\sqrt{3} + dt\,(F1 + F2)}{2\,dt} \tag{4}$$

```
> int1:=collect(simplify(int(linear_fit, x=0..t1)), {F1,F2});
```

$$int1 := \frac{F1\,dt}{4} + \frac{dt\left(-2\sqrt{3}+3\right)F2}{12} \tag{5}$$

```
> simplify((1/4-sqrt(3)/6)-(-2*sqrt(3) + 3)/12); # Confirm this is
  the same as the formula in the RK scheme
```

$$0 \tag{6}$$

```
> int2:=collect(simplify(int(linear_fit, x=0..t2)), {F1,F2});
```

$$int2 := \frac{dt\left(2\sqrt{3}+3\right)F1}{12} + \frac{F2\,dt}{4} \tag{7}$$

Write GaussRK2 scheme for a general f

So the GaussRK2 scheme is:

```
> F1:=f(X1,t1);
```

$$F1 := f\left(X1, \left(\frac{1}{2} - \frac{\sqrt{3}}{6}\right) dt\right) \tag{8}$$

```
> F2:=f(X2,t2);
```

$$F2 := f\left(X2, \left(\frac{1}{2} + \frac{\sqrt{3}}{6}\right) dt\right) \tag{9}$$

```
> eq1:=X1=X0+int1; # First stage is first equation
```

$$eq1 := X1 = X0 + \frac{f\left(X1, \left(\frac{1}{2} - \frac{\sqrt{3}}{6}\right) dt\right) dt}{4} + \frac{dt\left(-2\sqrt{3}+3\right) f\left(X2, \left(\frac{1}{2} + \frac{\sqrt{3}}{6}\right) dt\right)}{12} \tag{10}$$

```
> eq2:=X2=X0+int2; # Second stage is second equation
```

$$eq2 := X2 = X0 + \frac{dt\left(2\sqrt{3}+3\right) f\left(X1, \left(\frac{1}{2}-\frac{\sqrt{3}}{6}\right) dt\right)}{12} + \frac{f\left(X2, \left(\frac{1}{2}+\frac{\sqrt{3}}{6}\right) dt\right) dt}{4} \tag{11}$$

```
> system_eqs:={eq1, eq2}: # System of equations with unknowns X1
  and X2
```

Let's take a linear ODE

```
> f:=x->A*x:
> system_eqs;
```

$$\left\{X1 = X0 + \frac{A\, X1\, dt}{4} + \frac{dt\left(-2\sqrt{3}+3\right) A\, X2}{12}, X2 = X0 + \frac{dt\left(2\sqrt{3}+3\right) A\, X1}{12} + \frac{A\, X2\, dt}{4}\right\} \tag{12}$$

```
> sol:=solve(system_eqs,{X1,X2}); # Solve the system
```

$$sol := \left\{X1 = -\frac{2\sqrt{3}\, X0\left(A\, dt - 2\sqrt{3}\right)}{A^2\, dt^2 - 6\, A\, dt + 12}, X2 = \frac{2\sqrt{3}\, X0\left(A\, dt + 2\sqrt{3}\right)}{A^2\, dt^2 - 6\, A\, dt + 12}\right\} \tag{13}$$

```
> assign(sol):
> dX:=simplify(dt/2*(F1+F2)); # x^{n+1)-x^{n} for the GaussRK2
  scheme
```

$$dX := \frac{12\, dt\, A\, X0}{A^2\, dt^2 - 6\, A\, dt + 12} \tag{14}$$

```
> GaussRK2:=12*x/(x^2-6*x+12); # GaussRK2 approximation of exp(x)
```

$$GaussRK2 := \frac{12\, x}{x^2 - 6\, x + 12} \tag{15}$$

Confirm that this is a fourth order approximation of exp(x):

```
> series(GaussRK2,x,6); # Series expansion for GaussRK2 scheme
```

$$x + \frac{1}{2}\, x^2 + \frac{1}{6}\, x^3 + \frac{1}{24}\, x^4 + \frac{1}{144}\, x^5 + \mathrm{O}\left(x^6\right) \tag{16}$$

```
> series(exp(x),x,6);
```

$$1 + x + \frac{1}{2}\, x^2 + \frac{1}{6}\, x^3 + \frac{1}{24}\, x^4 + \frac{1}{120}\, x^5 + \mathrm{O}\left(x^6\right) \tag{17}$$